=====DC_Controlled_AllPass_Analog=====

```
*
*                                ______| b2 |___________________
*                               |      |____|                   |
*                               |                 ____| b1 |__________   |
*                               |Highpass        |    |____|          |  |
* U(S)                 ___    ..                 |  .                 V_V
*                     / _ \   y(t)     ____      |  y(t)  ____   y(t)  / _ \
*  _\| k |_\/ \      /____|_| 1/s |_|______| 1/s |____\/ \      \___\ Y(S)
*   /|___|  /\ \_/  /       |_____|         |_____|     /\ /_   /    /
*                \___/                                      \___/
*                 ^ ^              Bandpass
*                 | |___| -a2 |__________|                  b0 = 1
*                 |     |_____|             Lowpass
*                 |_______________| -a1 |_______________|    w0 = sqrt(a1)
*                                 |_____|                    Q  = w0/a2
*
*          Y(S)/U(S)  =   K*(1+ s*b1 + s^2*b2)/( s^2 +a2*s +a1)
*          Complex_Poles  1/(s^2 +a2*s +a1)   = 1/(s^2 +(wo/Q)*s +wo^2)
*.OPTIONS GMIN=1p METHOD=TRAP ABSTOL=1u TEMP=27 srcsteps = 1
```

Being able to DC program a universal state variable filter into a allpass filter comes in handy. To start off, the gain of the filter will be one regardless of frequency.






```
VFC1      FC1    0    DC     1
VK        K      0    DC     1
VA1       A1     0    DC     1
VA2       A2     0    DC     1
VB0       B0     0    DC     1
VB1       B1     0    DC     -1
VB2       B2     0    DC     1

XStateS1  VIN    FC1    K     A1  A2  B0  B1  B2  VOUT1  HP   BP   LP   StateVS
```

The DC voltages will all be one except for B1 which is set to minus one. The FC1 DC voltage defines where the phase shift happens. This subcircuit has been scaled such at a 1Hz signal will have 90 degrees of phase shift.

```
VFC1      FC1    0    DC     1
VFC2      FC2    0    DC     3
BVFC3     FC3    0    V      = V(FC2)*V(FC2)

XStateS1  VIN    FC1    K   A1  A2  B0  B1  B2  VOUT1  HP   BP   LP   StateVS
XStateS2  VIN    FC2    K   A1  A2  B0  B1  B2  VOUT2  HP2  BP2  LP2  StateVS
XStateS3  VIN    FC3    K   A1  A2  B0  B1  B2  VOUT3  HP3  BP3  LP3  StateVS
```

A single all pass has a nonlinear phase response. But what happens when 3 all pass filters are connected in series, and are frequency spaced by a factor of 3.





There will be some nonlinearity cancelation happening between the three all pass filters. The closer these filters are frequency spaced apart, the better the cancelation.

alter vfc2 **dc = 4**

The DC control voltages make it easy to use an alter statement to adjust the frequency spacing.





Increasing the frequency spacing does degrade the nonlinearity cancelation, but it also extends the linearity over a greater frequency range.

```
===========Full_Netlist_For_Copy_Paste=======================
Simple_StateVariable_Cell
*
*                            ___
*                     |_____| b2|___________________
*                     |     |___|                   |
*                     |                 ___         |
*                     |            ____| b1|_______ |
*                     |Highpass   |    |___|      | |
*  U(S)                           |                V_V
*             ___       ..        |  .            /  _  \
*        ___ / _ \     y(t)  ___     y(t)  ___  y(t) /  \     \
*  _\|  k  |_\/ \   \______|_| 1/s |_|______| 1/s |____\/ \    \___\ Y(S)
*   /|_____|/\ /_   /        |_____|    |   |_____| |   /\ /_    /    /
*            \___/                      |           |     \___/
*             ^  ^       Bandpass       |           |
*             |  |___| -a2 |____________|           |  b0 = 1
*             |      |_____|       Lowpass          |
*             |___________________| -a1 |___________|   w0 = sqrt(a1)
*                                 |_____|               Q  = w0/a2
*
*         Y(S)/U(S)   =   K*(1+ s*b1 + s^2*b2)/( s^2 +a2*s +a1)
*         Complex_Poles  1/(s^2 +a2*s +a1)  = 1/(s^2 +(wo/Q)*s +wo^2)
*
*.OPTIONS GMIN=1p METHOD=TRAP ABSTOL=1u TEMP=27 srcsteps = 1
*.OPTIONS RELTOL=.001 ABSTOL=1n VNTOL=1n ITL4=500
.OPTIONS GMIN=1p  srcsteps = 1 ITL1=4000 gminsteps = 1

*=========Create_Signal==================
VIN       VIN    0      DC     0    AC    1
VFC1      FC1    0      DC     1
VFC2      FC2    0      DC     3
BVFC3     FC3    0      V      = V(FC2)*V(FC2)
VK        K      0      DC     1
VA1       A1     0      DC     1
VA2       A2     0      DC     1
VB0       B0     0      DC     1
VB1       B1     0      DC     -1
VB2       B2     0      DC     1
XStateS1  VIN    FC1    K      A1  A2  B0  B1  B2  VOUT1  HP   BP   LP   StateVS
XStateS2  VIN    FC2    K      A1  A2  B0  B1  B2  VOUT2  HP2  BP2  LP2  StateVS
XStateS3  VIN    FC3    K      A1  A2  B0  B1  B2  VOUT3  HP3  BP3  LP3  StateVS
```

```
.control
set       pensize = 2
*AC       DECLin NUMDEC FSTART FSTOP
ac        dec    100    .3     30
let pht  = unwrap(ph(vout1)) + unwrap(ph(vout2))  + unwrap(ph(vout3))

let lin  = -640*log(frequency ) -235
plot      db(vout1) ylimit -1 1            title gain_Flatness
plot      db(vout1) unwrap(ph(vout1))      title AllPass
plot      db(vout1) unwrap(ph(vout1)) unwrap(ph(vout2)) unwrap(ph(vout3)) pht  lin        title AllPassOut
plot      pht - lin          title Fshift_3    ylimit -20 20

alter     vfc2         dc = 4
ac        dec    100    .3     30
let pht  = unwrap(ph(vout1)) + unwrap(ph(vout2))  + unwrap(ph(vout3))

let lin  = -535*log(frequency ) -217
plot      db(vout1) unwrap(ph(vout1)) unwrap(ph(vout2)) unwrap(ph(vout3)) pht  lin        title AllPassOut
plot      pht - lin          title Fshift_4  ylimit -20 20
.endc


*=========StateVariable_Cell_S==========================================
*
*
*  ___    |\  K      HP   |\  b2 gmb2                          ___
*| IN |__| \/\/\_______| \/\/\________________________________|OUT|
*|___|   | /\/\/       |  /\/\/                          |    |___|
*        |/  gmin      ||/       |\  b1 gmb1             |
*                      |        __| \/\/\________________|
*                      |       |  |  /\/\/               |
*                      |  Fscale  |/   Fscale            |
*       R1 =1          |  __|__   |    __|__   |\  b0    |      R2 =1
*     _/\  /\  /\___|_| 1/s  |_|_____| 1/s  |___| \/\/\_|_/\  /\  /\_
*    _|_ \/  \/     |  |_____|   |   |_____|  |  /\/\/    \/  \/  _|_
*    ///            |  s1      /|   |    s2   |  |/  gmb0          ///
*                   |_____/\/\/ |_|BP         |
*                          \/\/\ |gma1         |
*                         -a2  \|          /| |LP
*                   |______________________/\/\/ |_|
*                                       \/\/\ |gma2
*                                       -a1  \|
.SUBCKT   StateVS VIN    FC   K   A1  A2  B0  B1  B2  VOUT  HP  BP  LP
 R1        HP     0      1
 R2        OUT    0      1
 Bgmin     HP     0      I  = -V(VIN)*V(K)*1
 Bgma1     HP     0      I  =  V(LP)*V(A1)
 Bgma2     HP     0      I  =  (V(BP))*V(A2)
 Bgmb0     OUT    0      I  = -V(LP)*V(B0)
 Bgmb1     OUT    0      I  = -V(BP)*V(B1)
 Bgmb2     OUT    0      I  = -V(HP)*V(B2)
 XS1block  HP     BP     FC    Sblock
 XS2block  BP     LP     FC    Sblock
 BOUT      VOUT   0      V  =  V(OUT)
.ENDS     StateVS

*=========S_BLOCK========================================
*                           _______
*         IN   RS=1/FC  _|_       |
*        |\             ___       |
*|VIN|__| \_/\  /\  /\___|___|\   |  ___
*|___|   | /   \/  \/   VNEG  | \_|_|OUT|
*        |/Gain=-1        ____| /   |___|
*                        _|_  |/
*   For FC = 1Hz         ///
*   RS = 1 Ohm                CS=1uF/2Pi
*   Xc = 1 Ohm
.SUBCKT   Sblock VIN    OUT    FC
 Bbuf     IN     0      V =    -V(VIN)
 BRS      IN     VNEG   I =    (V(IN)-V(VNEG))*V(FC)
 Cs       VNEG   OUT    .159
 BSOUT    OUT    0      V =    -V(VNEG)*3000
.ENDS     Sblock

.end
```


6.7.11_12.30PM
dsauersanjose@aol.com
Don Sauer